# Initialization

Things like /. kxy2κList sometimes don't do what I want - for example, using κ+Pi instead of κ in plot arguments doesn't shift the lines, because kxy2κList remains the same. Define a function if this is the case.

In[1]:=
```
On[Assert]
```

# Constants

## Lattice vectors

I follow Tom's convention on vectors. I also set the lattice constant a = 1. The "I" and "II" means basis I & II (see my notes).

In[2]:=
```
exp23 = Exp[-2 * Pi * I / 3];
a1vec = {0, 1};
a2vec = {-Sqrt[3] / 2, 1 / 2};
a3vec = {Sqrt[3] / 2, 1 / 2};
ToLatVecList = {a1 → a1vec, a2 → a2vec, a3 → a3vec};
(*Transforming from basis II to I, if one cares*)
UII2I = {{1, 0, 0}, {0, Exp[I * k.a2], 0}, {0, 0, Exp[I * k.a3]}};
```

## Symmetry operators

These are written in the sublattice basis.

In[8]:=
```
C3II = {{0, 1, 0}, {0, 0, 1}, {1, 0, 0}};
MyII = {{0, 1, 0}, {1, 0, 0}, {0, 0, 1}};
ΠII = {{-1, 0, 0}, {0, -1, 0}, {0, 0, -1}};
```

## High symmetry basis states

In[11]:=
```
(*Eigensystem[C3II]*)
ψ0 = (1 / Sqrt[3]) * {1, 1, 1};
ψR = (1 / Sqrt[3]) * {exp23, 1, 1 / exp23};
ψL = (1 / Sqrt[3]) * {1 / exp23, 1, exp23};
```

MyII has a double root of 1, so we are free to form any linear superposition of them and still get an eigenstate. I chose the state such that they are also eigenstates for the lattice hamiltonian at Γ/K/K', respectively.

In[14]:=
```
Eigensystem[MyII]
(*ψ0 is the same as C3II*)
ψy1Γ = (1 / Sqrt[2]) * {-1, 1, 0};
ψy2Γ = (1 / Sqrt[6]) * {1, 1, -2};
ψy0K = (1 / Sqrt[2]) * {1, 0, 1};
ψy1K = (1 / Sqrt[2]) * {-1, 1, 0};
ψy2K = (1 / Sqrt[3]) * {1, 1, -1};
```

Out[14]=
```
{{-1, 1, 1}, {{-1, 1, 0}, {0, 0, 1}, {1, 1, 0}}}
```

# Kagome lattice - Expansion around Γ

## Constants

I will start with the chiral basis (the eigenstates of the C3 operator), and form linear superpositions of the chiral states.

In[20]:=
```
kNearΓ = d * {kx, ky}; (*d is a dummy variable for series expansion*)
kNearK = {d * kx, d * ky + Pi * 2 / 3};
kNearKb = {d * kx, d * ky - Pi * 2 / 3}; (*I'll write K' as Kb*)
kNearΓVecList = Append[ToLatVecList, k → kNearΓ];
kNearKVecList = Append[ToLatVecList, k → kNearK];
kNearKbVecList = Append[ToLatVecList, k → kNearKb];
kxy2κList = {kx → Cos[κ], ky → Sin[κ]};
kxy2ZeroList = {kx → 0, ky → 0};
```

### New basis states

$\psi$p and $\psi$m turn out to not be the nice basis that gives the reduced hamiltonian in Zephy's thesis. As we will see later, we want to use $\tau = \pi/3$ for the pseudospin basis. This turns out to be $\psi$y1Γ and $\psi$y2Γ - makes sense.

In[28]:=
```
ψp = Simplify[(ψR + ψL) / Sqrt[2]];
ψm = Simplify[(ψR - ψL) / (I * Sqrt[2])];
ψt1 = Simplify[(ψR + Exp[I * τ] * ψL) / Sqrt[2]];(*t for tau*)
ψt2 = Simplify[(ψR - Exp[I * τ] * ψL) / Sqrt[2] / I];
ψbs1 = Simplify[ψt1 /. τ → Pi / 3](*bs for Bloch sphere*)
ψbs2 = Simplify[ψt2 /. τ → Pi / 3]
Simplify[ψbs2[[3]] / ψbs2[[2]]]
```

Out[32]=
$$\left\{-\frac{1+(-1)^{1/3}}{\sqrt{6}}, \frac{1+(-1)^{1/3}}{\sqrt{6}}, 0\right\}$$

Out[33]=
$$\left\{\frac{-i+(-1)^{5/6}}{\sqrt{6}}, \frac{-i+(-1)^{5/6}}{\sqrt{6}}, (-1)^{1/6}\sqrt{\frac{2}{3}}\right\}$$

Out[34]=
$$-2$$

## Unitary matrices for basis transformation

In[35]:=
```
ULR = {ψ0, ψR, ψL};
Upm = {ψ0, ψp, ψm};
Utau = {ψ0, ψt1, ψt2};
Ubs = {ψ0, ψbs1, ψbs2};
```

# Kagome Hamiltonian

Convention for variable names (followed loosely):

- HkagII is in the sublattice basis (using basis II)
- HkagIItau is in the pseudospin basis defined by $\tau$
- HkagIIbs is in the "nicest" pseudospin basis. For the case of kagome lattice this is $\tau=\pi/3$

In[39]:=
```
HkagII =
  -2 * {{0, Cos[k.a1], Cos[k.a2]}, {Cos[k.a1], 0, Cos[k.a3]}, {Cos[k.a2], Cos[k.a3], 0}};
HkagIIatΓ = Simplify[HkagII /. k.a_ → 0];
HkagIIatK = Simplify[HkagII /. kNearKVecList /. kxy2ZeroList];
(*HkagIIatKb is the same as at K*)
HkagIItau = Utau.HkagII.Inverse[Utau];
HkagIItaupwr2 =
  Normal[Series[ComplexExpand[HkagIItau /. kNearΓVecList], {d, 0, 2}]] /. d → 1;
(*Expand[Simplify[TrigToExp[UII2I.HkagII.Inverse[UII2I]]]]//MatrixForm*)
```

In[44]:=
```
MatrixForm[HkagIIatΓ]
```

Out[44]//MatrixForm=
$$\begin{pmatrix} 0 & -2 & -2 \\ -2 & 0 & -2 \\ -2 & -2 & 0 \end{pmatrix}$$

## Sanity checks

At Γ: this should be diagonal and have a double root for the QBTP.

In[45]:=
```
HkagIItaupwr2atΓ = Simplify[HkagIItaupwr2 /. {kx → 0, ky → 0}];
MatrixForm[HkagIItaupwr2atΓ]
E0Pert = HkagIItaupwr2atΓ[[2, 2]];
(*E0Pert will be used in the 1st order perturbation below*)
```

Out[46]//MatrixForm=

$$\begin{pmatrix} -4 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{pmatrix}$$

The Hamiltonian at Γ should commute both C3 and My, but at K it only commutes with My:

In[48]:=
```
HkagIIatΓ.C3II - C3II.HkagIIatΓ
HkagIIatΓ.MyII - MyII.HkagIIatΓ
HkagIIatK.C3II - C3II.HkagIIatK(*Non-zero*)
HkagIIatK.MyII - MyII.HkagIIatK
```

Out[48]=
```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Out[49]=
```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Out[50]=
```
{{-2, 0, 2}, {0, 2, 0}, {0, -2, 0}}
```

Out[51]=
```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Full band structure looks right:

```
KagomeBands = Eigenvalues[Simplify[HkagII /. VecExprList /. {a → 1, d → 1}]];
kRange = Pi;
kSteps = 100;
kxVals = Subdivide[-kRange, kRange, kSteps];
kyVals = Subdivide[-kRange, kRange, kSteps];
energyGrids = Table[Table[Evaluate[b /. {kx → kxn, ky → kyn}],
    {kxn, kxVals}, {kyn, kyVals}], {b, KagomeBands}];
plots = Table[ListPlot3D[energyGrids[[i]], DataRange → {{-kRange, kRange}, {-kRange, kRange}},
    Mesh → None, PlotLabel → "Band " <> ToString[i],
    AxesLabel → {"kx", "ky", "E"}, ColorFunction → "Rainbow",
    Boxed → False, Lighting → "Neutral"], {i, Length[energyGrids]}];
Show[plots, PlotRange → All]
```

## Playing with $\tau$

$\tau$ -> Pi/3 gives the right basis for the pseudospin. Other $\tau$ doesn't look so nice.

In[52]:=
```
MatrixForm[Simplify[HkagIItaupwr2 /. τ → Pi / 3]](*Insert your favorite τ here*)
HkagIINicest = Simplify[HkagIItaupwr2 /. τ → Pi / 3];
```

Out[52]//MatrixForm=

$$\begin{pmatrix} -4+kx^2+ky^2 & \frac{\left(-i+\sqrt{3}\right)\,kx\,ky}{2\sqrt{2}} & \frac{\left(-i+\sqrt{3}\right)\left(kx^2-ky^2\right)}{4\sqrt{2}} \\ \frac{\left(i+\sqrt{3}\right)\,kx\,ky}{2\sqrt{2}} & 2-ky^2 & kx\,ky \\ \frac{\left(i+\sqrt{3}\right)\left(kx^2-ky^2\right)}{4\sqrt{2}} & kx\,ky & 2-kx^2 \end{pmatrix}$$

If we solve for the eigenvectors of the full "Nicest" Hamiltonian directly, we see that the ground state only changes at O(k^2) around Γ . Also, the admixtures of $\psi 0$ into the excited band states are also of order O(k^2). Compare this to the 0th order perturbation results below. (Probably should take the root warnings seriously)

```
(*The codes below shows the claim above, but takes some time to run*)
{evalsHkagIINicestFull, evecsHkagIINicestFullraw} =
  Simplify[Eigensystem[HkagIINicest], Assumptions → {kx, ky} ∈ Reals];
evecsHkagIINicestFull2ndraw = Simplify[
  Normal[Series[evecsHkagIINicestFullraw /. {kx → d * kx, ky → d * ky}, d → 0]] /. d → 1]
(*Normalized vectors. I was doing some calculations with them
 that didn't really work out, probably because of the root issue.
  I had to use simplify to get rid of most of the Abs functions,
so that when I take derivatives w.r.t. kx and
 ky below I don't get Abs' that can't be evaluated.*)
evecsHkagIINicestFull2nd =
  Simplify[Normalize[#] & /@ evecsHkagIINicestFull2ndraw, Assumptions → {kx, ky} ∈ Reals];
```

## The case of $\tau = \pi/3$ - mapping to pseudospins

We want to get a description of the two excited bands only, since the ground band is pretty decoupled from them. And by “pretty decoupled,” I mean that the off-diagonal terms in the Hamiltonian between the ground band and the excited bands are small in some sense. This “some sense” is made explicit by going to higher order perturbation.

### 0th order perturbation

This is obtained by trimming off the rows and columns involving the lowest band.

In[54]:=
```
HkagIIPspin0th = HkagIINicest[[2 ;; 3, 2 ;; 3]];
(*MatrixForm[HkagIINicest]*)
MatrixForm[HkagIIPspin0th]
```

Out[55]//MatrixForm=

$$\begin{pmatrix} 2-ky^2 & kx\,ky \\ kx\,ky & 2-kx^2 \end{pmatrix}$$

Pseudospin 1 and 2 from now on refer to the flatband states and the dispersive band states, respectively.

In[56]:=
```
{evals0th, evecs0thRaw} = Eigensystem[HkagIIPspin0th]
evecs0th = Simplify[Normalize[#] & /@ evecs0thRaw, Assumptions → {kx > 0, ky > 0}]
 (*Technically I should use {kx,ky}∈Reals instead of {kx>0,ky>0},
 but this only leaves an unneccary global phase in front of each eigenvector*)
```

Out[56]=

$$\left\{\left\{2, 2-kx^2-ky^2\right\}, \left\{\left\{\frac{kx}{ky}, 1\right\}, \left\{-\frac{ky}{kx}, 1\right\}\right\}\right\}$$

Out[57]=

$$\left\{\left\{\frac{kx}{\sqrt{kx^2+ky^2}}, \frac{ky}{\sqrt{kx^2+ky^2}}\right\}, \left\{-\frac{ky}{\sqrt{kx^2+ky^2}}, \frac{kx}{\sqrt{kx^2+ky^2}}\right\}\right\}$$

(Not the best way to plot this, but) we see that the state winds two times around the QBTP as expected:

In[58]:=
```
Plot[evecs0th[[1]][[#]] /. kxy2κList, {κ, 0, 2 Pi}] & /@ {1, 2}
```

Out[58]=




The eigenenergy (and therefore the band gap) on a circle around Γ is constant :

In[59]:=
```
Eigensystem[HkagIIPspin0th /. kxy2κList]
```

Out[59]=

```
{{2, 1}, {{Cot[κ], 1}, {Cot[κ] - Csc[κ] Sec[κ], 1}}}
```

## 1 st order perturbation - not needed

Since the off-diagonal terms are of order k^2, we should try to go to 1st order perturbation to get the correct quadratic matrix. However, one sees that the lowest order correction is actually of order k^4, so we can safely ignore this term. In hindsight this is very obvious.

In[60]:=
```
E3Pert = HkagIINicest[[1, 1]];
cPert = HkagIINicest[[1, 2]];
bPert = HkagIINicest[[1, 3]];
HPert1stWithd = Simplify[{{Abs[cPert]^2, cPert * bPert*}, {cPert* * bPert, Abs[bPert]^2}} /
      (E0Pert - E3Pert)] /. {kx → d * kx,  ky → d * ky};
HPert1st = Simplify[Normal[Series[ComplexExpand[HPert1stWithd], {d, 0, 4}]] /. d → 1,
   Assumptions → {kx ∈ Reals, ky ∈ Reals}];
MatrixForm[HPert1st]
```

Out[65]//MatrixForm=

$$\begin{pmatrix} \frac{kx^2\, ky^2}{12} & \frac{1}{24} kx\, ky \left(kx^2-ky^2\right) \\ \frac{1}{24} kx\, ky \left(kx^2-ky^2\right) & \frac{1}{48}\left(kx^2-ky^2\right)^2 \end{pmatrix}$$

# Shaking Hamiltonian

See my notes on how I got the perturbation term K (which is not Hermitian if Ξ is not purely real), and the definition of Ξ. I am not too confident on whether K is correct. In any case, K is periodic in $\kappa$ with a period of $2\pi$ (or $\pi$, since the overall sign gets squared out when finding the transition amplitude).

In[66]:=
```
KShakeKagII = {{0, Sin[k.a1] * (Ξ.a1), Sin[k.a2] * (Ξ.a2)},
   {Sin[k.a1] * (Ξ.a1), 0, Sin[k.a3] * (Ξ.a3)}, {Sin[k.a2] * (Ξ.a2), Sin[k.a3] * (Ξ.a3), 0}};
MatrixForm[KShakeKagII]
```

Out[67]//MatrixForm=

$$\begin{pmatrix} 0 & \Xi.a1\,\text{Sin}[k.a1] & \Xi.a2\,\text{Sin}[k.a2] \\ \Xi.a1\,\text{Sin}[k.a1] & 0 & \Xi.a3\,\text{Sin}[k.a3] \\ \Xi.a2\,\text{Sin}[k.a2] & \Xi.a3\,\text{Sin}[k.a3] & 0 \end{pmatrix}$$

## Linear shaking

### Matrix element

This is found by first finding the matrix element between the ground state and the pseudospin basis states, and then forming the superposition of the matrix elements. i.e.,
$\langle\kappa|\psi 0\rangle = (A\,\langle\psi bs1| + B\,\langle\psi bs2|)\,|\psi 0\rangle = A\,\langle\psi bs1|\psi 0\rangle + B\,\langle\psi bs2|\psi 0\rangle$.

In[68]:=
```
KmatPspin1Kag = FullSimplify[ψbs1.KShakeKagII.ψ0];
KmatPspin2Kag = FullSimplify[ψbs2.KShakeKagII.ψ0];
(*1 and 2 above are the two Bloch sphere basis states*)
(*1 and 2 below are the two excited bands*)
(*The ComplexExpand is pretty crucial for some reason*)
KmatKagNearΓ1 = Normal[Series[ComplexExpand[
      evecs0th[[1]].{KmatPspin1Kag, KmatPspin2Kag} /. kNearΓVecList], {d, 0, 2}]] /. d → 1;
KmatKagNearΓ2 = Normal[Series[ComplexExpand[
      evecs0th[[2]].{KmatPspin1Kag, KmatPspin2Kag} /. kNearΓVecList], {d, 0, 2}]] /. d → 1;
(*One can also choose to not expand to O(k),
which gives qualitatively identical results*)
(*KmatKagNearΓ1=
 ComplexExpand[evecs0th[[1]].{KmatPspin1Kag,KmatPspin2Kag}/.kNearΓVecList]/.d→1;
KmatKagNearΓ2=
 ComplexExpand[evecs0th[[2]].{KmatPspin1Kag,KmatPspin2Kag}/.kNearΓVecList]/.d→1;*)
```

In[72]:=
```
GetMatEleLinShake1 =
  Function[{κ, λ}, Simplify[KmatKagNearΓ1 /. kxy2κList /. Ξ → {Cos[λ], Sin[λ]}]];
GetExAmpLinShakePspin1 = Function[{κ, λ}, Simplify[Abs[GetMatEleLinShake1[κ, λ]]^2]];
GetMatEleLinShake2 =
  Function[{κ, λ}, Simplify[KmatKagNearΓ2 /. kxy2κList /. Ξ → {Cos[λ], Sin[λ]}]];
GetExAmpLinShakePspin2 = Function[{κ, λ}, Simplify[Abs[GetMatEleLinShake2[κ, λ]]^2]];
```

In[76]:=
```
(*Simplify[KmatKagNearΓ1,Assumptions→{κ,λ}∈Reals]*)
Simplify[GetExAmpLinShakePspin1[κ, λ], Assumptions → {κ, λ} ∈ Reals]
```

Out[76]=

$$\frac{1}{8}\,\text{Sin}[2\kappa+\lambda]^2$$

## Finding nodes of excitation

The solution can simply be written as $\frac{1}{2}(-\lambda+\pi m)$ for any integer n (or just n = 0,1,2,3). The factor of one half is interesting. I assume that this showed up because my perturbation and the state decomposition is k-dependent in a non-trivial way, but I don't really know for sure.

In[77]:=
```
LinShakeNodes = Reduce[GetExAmpLinShakePspin1[κ, λ] == 0, κ, Reals]
```

Out[77]=

$$\mathbb{c}_1 \in \mathbb{Z}\ \&\&\ \left(\kappa == -\frac{\lambda}{2}+\pi\,\mathbb{c}_1\ ||\ \kappa == -\frac{\lambda}{2}+\frac{1}{2}(\pi+2\pi\,\mathbb{c}_1)\right)$$

```
lambdaList = Pi * Range[0, 1, 0.1];
realSolsTable = Association[Table[λval = N[λ0];
    (*Ensure numerical*)sols = Solve[GetExAmpLinShakePspin1[κ, λval] == 0, κ];
    realKappa = Select[κ /. sols, Im[#] == 0 &];
    λ0 → realKappa, {λ0, lambdaList}]];
numsol =
  Flatten[KeyValueMap[Function[{λval, κlist}, Thread[{λval, κlist}]], realSolsTable], 1];
ListPlot[numsol, PlotStyle → PointSize[Medium],
 AxesLabel → {"λ", "κ"}, PlotRange → All, AspectRatio → 1, GridLines → Automatic,
 Frame → True, FrameLabel → {"λ", "κ"}, LabelStyle → {FontSize → 14}]
```

Out[•]=

$$\mathbb{c}_1 \in \mathbb{Z}\ \&\&\ \left(\kappa == -\frac{\lambda}{2}+\pi\,\mathbb{c}_1\ ||\ \kappa == -\frac{\lambda}{2}+\frac{1}{2}(\pi+2\pi\,\mathbb{c}_1)\right)$$

## Plots

A few things to notice:

1. $\lambda'=\lambda+\pi$ should be the same as $\lambda$, since they are actually shaking in the same direction.
2. The excitation pattern simply gets rotated around as one varies $\lambda$. In particular, this might mean that a winding number of +2 v.s. -2 can be distinguished based on how the pattern moves around. In the second plot, both traces move in the save direction as $\lambda$ is varied since they should sum to a constant value.
3. C3 symmetry is preserved - this is pretty hard to see (took me a whole afternoon to realize, but remember the minus sign in front of $\lambda$!)

In[78]:=
```
Plot[Evaluate@Table[GetExAmpLinShakePspin1[κ, λ], {λ, {0, 2 * Pi / 3, 4 * Pi / 3, Pi}}],
 {κ, -Pi, Pi},
 PlotLegends → Placed[LineLegend[{"λ = 0", "λ = 2π/3", "λ = 4π/3", "λ = π"}], Right],
 PlotStyle → {Red, Blue, Green, {Gray, Dashed}}]
```

Out[78]=




In[79]:=
```
Manipulate[Plot[{GetExAmpLinShakePspin1[κ, λ], GetExAmpLinShakePspin2[κ, λ]},
  {κ, -Pi, Pi}, PlotRange → {0, 0.15}, AxesLabel → {"κ", "Amplitude"},
  PlotLegends → Placed[LineLegend[{"Band 3 (flat)", "Band 2"}], Right]],
 {{λ, 0, "λ"}, 0, Pi, Appearance → "Labeled"}]
```

Out[79]=




## Circular shaking

We'll see that circular shaking doesn't yield any information on the geometry of the band structure.

### Matrix elements

In[80]:=
```
GetMatEleCircShake =
  Function[{κ}, Simplify[KmatKagNearΓ1 /. {Ξ → {I, 1}, kx → k * Cos[κ], ky → k * Sin[κ]}]];
GetExAmpCircShake = Function[{κ}, Simplify[Abs[GetMatEleCircShake[κ] /. k → 1]^2]];
FullSimplify[GetMatEleCircShake[κ], Assumptions → κ ∈ Reals]
FullSimplify[GetExAmpCircShake[κ], Assumptions → κ ∈ Reals]
```

Out[82]=

$$\frac{\left(\mathrm{i}+\sqrt{3}\right) e^{2\,\mathrm{i}\,\kappa}\sqrt{k^2}}{4\sqrt{2}}$$

Out[83]=

$$\frac{1}{8}$$

### Plots

In[84]:=
```
Plot[GetExAmpCircShake[κ], {κ, 0, 2 Pi}]
```

Out[84]=




## Matrix element, higher order, but wrong

# Kagome lattice - Expansion around K and K’

## Constants

I will use the eigenstates of the mirror symmetry operator My as the basis for the two lower bands.

In[85]:=
```
ψy1Γ
ψy2Γ
testmat = HkagII /. kNearKbVecList /. kxy2ZeroList
HkagIIatK
Simplify[Eigensystem[HkagIIatK]]
```

Out[85]=
$$\left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0\right\}$$

Out[86]=
$$\left\{\frac{1}{\sqrt{6}}, \frac{1}{\sqrt{6}}, -\sqrt{\frac{2}{3}}\right\}$$

Out[87]=
```
{{0, 1, -1}, {1, 0, -1}, {-1, -1, 0}}
```

Out[88]=
```
{{0, 1, -1}, {1, 0, -1}, {-1, -1, 0}}
```

Out[89]=
```
{{2, -1, -1}, {{-1, -1, 1}, {1, 0, 1}, {-1, 1, 0}}}
```

### New basis states

# Lieb-kagome lattice - expand near emerging Dirac points

## Constants

I will use the eigenstates of the mirror symmetry operator My as the basis for the excited bands. Note that they have opposite eigenvalues for My.

In[90]:=
```
rLk = 0.8;
Assert[1 ≥ rLk ≥ 0]
```

### New basis states

Since away from the high symmetry line we don't expect $\psi$y1Γ and $\psi$y2Γ to be eigenstates of the hamiltonian, but they should be on the high symmetry line, the weight between $\psi$y1Γ and $\psi$y2Γ should vary as a function of angle. I'll similarly guess that the nice basis for our Bloch sphere picture will be given by the superposition of these two states with equal weight and perhaps some non-trivial relative phase.

In[92]:=
```
ψyt1 = Simplify[(ψy1Γ + Exp[I * τ] * ψy2Γ) / Sqrt[2]];
ψyt2 = Simplify[(ψy1Γ - Exp[I * τ] * ψy2Γ) / Sqrt[2]];
ψybs1 = ψyt1 /. τ → 0;
ψybs2 = ψyt2 /. τ → 0;
```

### Unitary matrices for basis transformation

In[96]:=
```
Uytau = {ψ0, ψyt1, ψyt2};
Uybs = {ψ0, ψybs1, ψybs2};
```

---

# Lieb-kagome lattice Hamiltonian

In[98]:=
```
HLkII = -2 *
   {{0, r * Cos[k.a1], Cos[k.a2]}, {r * Cos[k.a1], 0, Cos[k.a3]}, {Cos[k.a2], Cos[k.a3], 0}};
```

### Sanity checks

It is easy to see that My should commute with the Hamiltonian for kx = 0. But since H(k) = H(-k), the commutation relation also holds for ky = 0:

In[99]:=
```
HLkIIalongky = HLkII /. kNearΓVecList /. {kx → 0, d → 1};
HLkIIalongkx = HLkII /. kNearΓVecList /. {ky → 0, d → 1};
HLkIIalongky.MyII - MyII.HLkIIalongky
HLkIIalongkx.MyII - MyII.HLkIIalongkx
```

Out[101]=
```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Out[102]=
```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

### Position of the Dirac points

Note that the Dirac point split from the QBTP is spawn along the kx-axis. The bands were not ordered in energy. Also, while there appears to be a completely flat band, it is only flat along this high symmetry line.

In[103]:=
```
(*r→rLK must be applied after the Simplify below*)
{evalsHLkalongkx, evecsHLkalongkx} =
  Simplify[Eigensystem[HLkIIalongkx /. d → 1], Assumptions → 1 ≥ r ≥ 0];
```

Let's visualize this at the value of r chosen:

In[104]:=
```
Plot[Evaluate[evalsHLkalongkx /. r → rLk], {kx, -Pi, Pi}, PlotLegends → {"1", "2", "3"}]
{LkDiracPtkx1, LkDiracPtkx2} = Solve[evalsHLkalongkx[[1]] - evalsHLkalongkx[[3]] == 0, kx]
kNearLkDP1 = {d * kx - Values[LkDiracPtkx1][[1]], d * ky};
kNearLkDP2 = {d * kx - Values[LkDiracPtkx2][[1]], d * ky};
(*d is a dummy variable for series expansion*)
Lk1VecList = Append[ToLatVecList, k → kNearLkDP1];
Lk2VecList = Append[ToLatVecList, k → kNearLkDP2];
```

Out[104]=




Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[105]=

$$\left\{\left\{kx \to -\frac{\text{ArcCos}\left[-1+2r^2\right]}{\sqrt{3}}\right\}, \left\{kx \to \frac{\text{ArcCos}\left[-1+2r^2\right]}{\sqrt{3}}\right\}\right\}$$

## Expansion around the Dirac points

We only expand to lowest order in k.

In[110]:=
```
HLkIItau = Uytau.HLkII.Inverse[Uytau];
HLkIIDP1tauPwr1 = FullSimplify[
   Normal[Series[HLkIItau /. Lk1VecList, {d, 0, 1}]] /. d → 1, Assumptions → 1 ≥ r ≥ 0];
HLkIIDP2tauPwr1 = FullSimplify[Normal[Series[HLkIItau /. Lk2VecList, {d, 0, 1}]] /. d → 1,
   Assumptions → 1 ≥ r ≥ 0];(*Much better than Simplify in this case*)
```

## Playing around with $\tau$

$\tau$=0 seems good. Other choices will leave both kx and ky in the off-diagonal terms in the effective two-band hamiltonian.

In[113]:=
```
MatrixForm[Simplify[HLkIIDP1tauPwr1 /. τ → 0]](*Insert your favorite τ here*)
(*MatrixForm[Simplify[HLkIIDP2tauPwr1/.τ→0]](*Insert your favorite τ here*)*)
HLkIIDP1Nicest = Simplify[HLkIIDP1tauPwr1 /. τ → 0];
HLkIIDP2Nicest = Simplify[HLkIIDP2tauPwr1 /. τ → 0];
```

Out[113]//MatrixForm=

$$\begin{pmatrix} -\frac{4}{3}\left(3r - kx\sqrt{3-3r^2}\right) & \frac{1}{3}(-kx+ky)\sqrt{3-3r^2} & \frac{1}{3}(kx+ky)\sqrt{3-3r^2} \\ \frac{1}{3}(-kx+ky)\sqrt{3-3r^2} & 2r - \frac{2}{3}(kx+ky)\sqrt{3-3r^2} & \frac{2}{3}kx\sqrt{3-3r^2} \\ \frac{1}{3}(kx+ky)\sqrt{3-3r^2} & \frac{2}{3}kx\sqrt{3-3r^2} & 2r + \frac{2}{3}(-kx+ky)\sqrt{3-3r^2} \end{pmatrix}$$

This is really weird - I thought I'll get My symmetry, i.e. the first should be zero but not the second

In[116]:=
```
Simplify[HLkIIDP1Nicest - (HLkIIDP2Nicest /. {kx → -kx, ky → ky})]
Simplify[HLkIIDP1Nicest - (HLkIIDP2Nicest /. {kx → -kx, ky → -ky})]
```

Out[116]=

$$\left\{\left\{0, \frac{2ky\sqrt{1-r^2}}{\sqrt{3}}, \frac{2ky\sqrt{1-r^2}}{\sqrt{3}}\right\}, \left\{\frac{2ky\sqrt{1-r^2}}{\sqrt{3}}, -\frac{4ky\sqrt{1-r^2}}{\sqrt{3}}, 0\right\}, \left\{\frac{2ky\sqrt{1-r^2}}{\sqrt{3}}, 0, \frac{4ky\sqrt{1-r^2}}{\sqrt{3}}\right\}\right\}$$

Out[117]=
```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

## The case of $\tau = 0$

### 0th order perturbation

In[118]:=
```
HLkDP1Pspin0th = HLkIIDP1Nicest[[2 ;; 3, 2 ;; 3]];
HLkDP2Pspin0th = HLkIIDP2Nicest[[2 ;; 3, 2 ;; 3]];
(*MatrixForm[HLkIIDP1Nicest]*)
MatrixForm[HLkDP1Pspin0th]
```

Out[120]//MatrixForm=

$$\begin{pmatrix} 2r - \frac{2}{3}(kx+ky)\sqrt{3-3r^2} & \frac{2}{3}kx\sqrt{3-3r^2} \\ \frac{2}{3}kx\sqrt{3-3r^2} & 2r + \frac{2}{3}(-kx+ky)\sqrt{3-3r^2} \end{pmatrix}$$

Interestingly, one finds that the eigenstates around the Dirac points are independent of r, although the eigenvalues does:

In[121]:=
```
{evalsHLkDP1Pspin0th, evecsHLkDP1Pspin0thRaw} =
 Simplify[Eigensystem[HLkDP1Pspin0th], Assumptions → {{kx, ky} ∈ Reals, 1 ≥ r ≥ 0}]
evecsHLkDP1Pspin0th =
  Simplify[Normalize[#] & /@ evecsHLkDP1Pspin0thRaw, Assumptions → {kx, ky} ∈ Reals];
{evalsHLkDP2Pspin0th, evecsHLkDP2Pspin0thRaw} =
  Simplify[Eigensystem[HLkDP2Pspin0th], Assumptions → {{kx, ky} ∈ Reals, 1 ≥ r ≥ 0}];
evecsHLkDP2Pspin0th =
  Simplify[Normalize[#] & /@ evecsHLkDP2Pspin0thRaw, Assumptions → {kx, ky} ∈ Reals];
```

Out[121]=
$$\left\{\left\{-\frac{2}{3}\left(-3r+kx\sqrt{3-3r^2}+\sqrt{3}\sqrt{\left(kx^2+ky^2\right)\left(1-r^2\right)}\right),\ 2r+\frac{2\left(-kx\sqrt{1-r^2}+\sqrt{-\left(\left(kx^2+ky^2\right)\left(-1+r^2\right)\right)}\right)}{\sqrt{3}}\right\},\right.$$
$$\left.\left\{\left\{-\frac{ky+\sqrt{kx^2+ky^2}}{kx},\ 1\right\},\ \left\{\frac{-ky+\sqrt{kx^2+ky^2}}{kx},\ 1\right\}\right\}\right\}$$

### 1st order perturbation - not needed for linear touching

This is again obvious from the fact that the off-diagonal terms that couples to the ground band is linear order (and have no constant terms, i.e. is homogeneous).

---

# Shaking Hamiltonian

In[125]:=
```
KshakeLk = {{0, r * Sin[k.a1] * (Ξ.a1), Sin[k.a2] * (Ξ.a2)}, {r * Sin[k.a1] * (Ξ.a1),
    0, Sin[k.a3] * (Ξ.a3)}, {Sin[k.a2] * (Ξ.a2), Sin[k.a3] * (Ξ.a3), 0}};
MatrixForm[KshakeLk]
```

Out[126]//MatrixForm=
$$\begin{pmatrix} 0 & r\Xi.a1\,\text{Sin}[k.a1] & \Xi.a2\,\text{Sin}[k.a2] \\ r\Xi.a1\,\text{Sin}[k.a1] & 0 & \Xi.a3\,\text{Sin}[k.a3] \\ \Xi.a2\,\text{Sin}[k.a2] & \Xi.a3\,\text{Sin}[k.a3] & 0 \end{pmatrix}$$

## Linear shaking

### Matrix elements

Even though we are no longer centered around a high symmetry point, I still assume that the ground band state doesn't change too much.

In[127]:=
```
(*Simplify in the next cell doesn't make this simplification,
but using FullSimplify makes the execution too long*)
(*FullSimplify[Cos[1/2 ArcCos[-1+2 r^2]],Assumptions→1≥r≥0]
 FullSimplify[Sin[1/2 ArcCos[-1+2 r^2]],Assumptions→1≥r≥0]*)
simplifyHelperList = {Cos[1/2 ArcCos[-1 + 2 r^2]] → r, Sin[1/2 ArcCos[-1 + 2 r^2]] → Sqrt[1 - r^2]};
```

In[128]:=
```
KmatPspin1Lk = FullSimplify[ψybs1.KshakeLk.ψ0];
KmatPspin2Lk = FullSimplify[ψybs2.KshakeLk.ψ0];
(*1 and 2 above are the two Bloch sphere basis states*)
(*1 and 2 below are the two Dirac points of the same band*)
KmatLkNearDP1 = Simplify[
   Normal[Series[evecsHLkDP1Pspin0th[[1]].{KmatPspin1Lk, KmatPspin2Lk} /. Lk1VecList,
       {d, 0, 2}]] /. d → 1 /. simplifyHelperList, Assumptions → {{kx, ky} ∈ Reals, 1 ≥ r ≥ 0}];
KmatLkNearDP2 = Simplify[
   Normal[Series[evecsHLkDP2Pspin0th[[1]].{KmatPspin1Lk, KmatPspin2Lk} /. Lk2VecList,
       {d, 0, 2}]] /. d → 1 /. simplifyHelperList, Assumptions → {{kx, ky} ∈ Reals, 1 ≥ r ≥ 0}];
```

In[132]:=
```
fSimpHelperList = {Pi ≥ κ ≥ -Pi, Pi ≥ λ ≥ -Pi, 1 ≥ r ≥ 0};
GetLkDP1MatEleLinShake = Function[{κ, λ}, FullSimplify[
    KmatLkNearDP1 /. kxy2κList /. ⊠ → {Cos[λ], Sin[λ]}, Assumptions → fSimpHelperList]];
GetLkDP1ExAmpLinShake = Function[{κ, λ, rr}, FullSimplify[
    Abs[GetLkDP1MatEleLinShake[κ, λ] /. r → rr]^2, Assumptions → fSimpHelperList]];
GetLkDP2MatEleLinShake = Function[{κ, λ}, FullSimplify[
    KmatLkNearDP2 /. kxy2κList /. ⊠ → {Cos[λ], Sin[λ]}, Assumptions → fSimpHelperList]];
GetLkDP2ExAmpLinShake = Function[{κ, λ, rr}, FullSimplify[
    Abs[GetLkDP2MatEleLinShake[κ, λ] /. r → rr]^2, Assumptions → fSimpHelperList]];
```

## Position of nodes

The position of nodes is pretty nontrivial. Even the numerical solver seems to need some help to remove some prefactors that won't affect the position of nodes. Very interestingly there are three nodes.

In[137]:=
```
(*This term appears during simplification and is never zero*)
Reduce[Sec[1/4 (π + 2 κ)] √(1 - Sin[κ]) == 0, κ, Reals]
```

Out[137]=
```
False
```

In[138]:=
```
Step1LkDP1MatEle = FullSimplify[
   TrigFactor[GetLkDP1MatEleLinShake[κ, λ, r]], Assumptions → fSimpHelperList];
Step2LkDP1MatEle = Step1LkDP1MatEle / (1/(16 √3) Sec[1/4 (π + 2 κ)] √(1 - Sin[κ]))
```

Out[139]=
$$4\sqrt{3}\, r \cos\left[\frac{3\kappa}{2}+\lambda\right]+\sqrt{1-r^2}\left(-\cos\left[\frac{5\kappa}{2}-\lambda\right]+6\cos\left[\frac{\kappa}{2}+\lambda\right]+2\sin\left[\frac{\kappa}{2}\right]\sin[2\kappa+\lambda]\right)$$

In[140]:=
```
(*Reduce[Step2LkDP1MatEle==0,κ,Reals]*)
lambdaList = Pi * Range[-1, 1, 0.1];
LHSrealSol = Function[{κin, λin}, Step2LkDP1MatEle /. {r → rLk, κ → κin, λ → λin}];
realSolsTable = Association[Table[λval = N[λ0];
    (*Ensure numerical*)sols = Solve[LHSrealSol[κ, λval] == 0, κ];
    realKappa = Select[κ /. sols, Im[#] == 0 &];
    λ0 → realKappa, {λ0, lambdaList}]];
numsol =
  Flatten[KeyValueMap[Function[{λval, κlist}, Thread[{λval, κlist}]], realSolsTable], 1];
ListPlot[numsol, PlotStyle → PointSize[Medium], AxesLabel → {"λ", "κ"},
 PlotRange → {-Pi, Pi}, AspectRatio → 1, GridLines → Automatic,
 Frame → True, FrameLabel → {"λ", "κ"}, LabelStyle → {FontSize → 14}]
(*Manipulate[Plot[Step2LkDP1MatEle/.r→rLk/.λ→λλ,{κ,-Pi,Pi},PlotRange→{-10,10}],
 {{λλ,0,"λ"},0,Pi,Appearance→"Labeled"}]*)
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

General: Further output of Solve::ifun will be suppressed during this calculation.

Out[144]=




## Plots

xy-plot is not so useful in visualizing the excitation pattern, but is good for getting a feeling of the actual values of the matrix elements.

In[145]:=

```
Manipulate[Plot[{GetLkDP1ExAmpLinShake[κ, λ, rLk], GetLkDP2ExAmpLinShake[κ, λ, rLk]},
  {κ, -Pi, Pi}, PlotRange → {0, 0.3}, AxesLabel → {"κ", "Amplitude"},
  PlotLegends → Placed[LineLegend[{"DP1", "DP2"}], Right]],
 {{λ, 0, "λ"}, 0, Pi, Appearance → "Labeled"}]
```

Out[145]=




Both patterns move counterclockwise as $\lambda$ is increased, which is a result of the weird symmetry above. Is this telling us that the winding number around each of the Dirac point is indeed of the same sign? If so, the “weird” symmetry is actually expected

In[146]:=
```
colorFunc[val_] := Blend[{Blue, Red}, Rescale[Sqrt[val], {0, 0.5}]]
λPrmPlot = 0;
CircLkDP1ExAmpLinShake = ParametricPlot[{-1.1, 0} + {Cos[κ], Sin[κ]}, {κ, -Pi, Pi},
   ColorFunction → Function[{x, y, κ}, colorFunc[GetLkDP1ExAmpLinShake[κ, λPrmPlot, rLk]]],
   ColorFunctionScaling → False];
CircLkDP2ExAmpLinShake = ParametricPlot[{1.1, 0} + {Cos[κ], Sin[κ]} , {κ, -Pi, Pi},
   ColorFunction → Function[{x, y, κ}, colorFunc[GetLkDP2ExAmpLinShake[κ, λPrmPlot, rLk]]],
   ColorFunctionScaling → False];
Show[CircLkDP1ExAmpLinShake, CircLkDP2ExAmpLinShake,
 PlotRange → All, PlotLabel → "Red is more excitation"]
```

Out[150]=




## Circular shaking

This doesn't seem to tell us much.

In[151]:=
```
GetLkDP1MatEleCircShake = Function[{κ, rin}, FullSimplify[
     KmatLkNearDP1 /. kxy2κList /. ⊠ → {I, 1} /. r → rin, Assumptions → fSimpHelperList]];
GetLkDP1ExAmpCircShake = Function[{κ, rin},
   FullSimplify[Abs[GetLkDP1MatEleCircShake[κ, rin]]^2, Assumptions → fSimpHelperList]];
GetLkDP2MatEleCircShake = Function[{κ, rin}, FullSimplify[
     KmatLkNearDP2 /. kxy2κList /. ⊠ → {I, 1} /. r → rin, Assumptions → fSimpHelperList]];
GetLkDP2ExAmpCircShake = Function[{κ, rin},
   FullSimplify[Abs[GetLkDP2MatEleCircShake[κ, rin]]^2, Assumptions → fSimpHelperList]];
```

## Plots

In[155]:=
```
Plot[{GetLkDP1ExAmpCircShake[κ, rLk], GetLkDP2ExAmpCircShake[κ, rLk]}, {κ, 0, 2 Pi}]
```

Out[155]=




In[156]:=
```
CircLkDP1ExAmpCircShake = ParametricPlot[{-1.1, 0} + {Cos[κ], Sin[κ]}, {κ, -Pi, Pi},
    ColorFunction → Function[{x, y, κ}, colorFunc[GetLkDP1ExAmpCircShake[κ, rLk]]],
    ColorFunctionScaling → False];
CircLkDP2ExAmpCircShake = ParametricPlot[{1.1, 0} + {Cos[κ], Sin[κ]} , {κ, -Pi, Pi},
    ColorFunction → Function[{x, y, κ}, colorFunc[GetLkDP2ExAmpCircShake[κ, rLk]]],
    ColorFunctionScaling → False];
Show[CircLkDP1ExAmpCircShake, CircLkDP2ExAmpCircShake,
 PlotRange → All, PlotLabel → "Red is more excitation"]
```

Out[158]=